UNIVERSIDADE FEDERAL DO RIO GRANDE DO SUL

INSTITUTO DE MATEMÁTICA
DEPARTAMENTO DE MATEMÁTICA PURA E APLICADA
Disciplina: MAT01168 -Matemática Aplicada -Semestre Letivo 2008/2
Professoras: Elisabeta Gallicchio e Irene Strauch
TERCEIRA ÁREA
TRANSFORMADA DE FOURIER UTILIZANDO O MAPLE

Décima primeira aula 24/11/2008 **Propriedades da Transformada de Fourier**

Aplicações da segunda propriedade: Transformada da derivada no domínio tempo

Exemplos de resolução de E.D. Ordinárias com a transformada de Fourier

Exemplo 1 Não são dadas as condições iniciais

```
> restart:with(inttrans):with(plots):
> edo1 := diff(y(t),t$2) - 4*y(t)=4*Heaviside(t);
> sol1:=dsolve(edo1,  y(t),method = fourier);
> solexp:=convert(sol1,exp);
> sol_piec:=convert(sol1,piecewise);
> plot(rhs(sol_piec),t=-10..10,thickness=2,discont=true,titlefont=
  [COURIER,DEFAULT,16],labels=["t", "sol"],labelfont=[COURIER,
  DEFAULT,16],axesfont=[COURIER,DEFAULT,16],title=`Solução da EDO`)
  ;
```

===

Exemplo 2

$$edo := \frac{d}{dt}\, y(t) - y(t) = e^{-t^2}$$

```
> restart;with(inttrans):with(plots):
> edo := diff(y(t),t)-y(t)=exp(-t^2);
```

Resolvendo diretamente a edo

```
> dsolve(edo,y(t),method = fourier);
```

Importante: note que a transformada da exponencial e^{-t^2} é a função erro denotada por **erf**

Resolvendo passo a passo.

Primeiro passo: aplica-se a transformada de Fourier à equação diferencial

```
> trans1:=fourier(edo,t,w);
```

Para visualizar melhor o resultado substitui-se $fourier(y(t), t, w)$=Y(w)

```
> trans1_mod:=subs(fourier(y(t),t,w)=Y(w),trans1);
```

Segundo passo : resolve-se a equação algébrica para isolar Y(w)

```
> Y(w):=solve(trans1_mod, Y(w));
```

Só por curiosidade, vamos separar os coeficientes da parte real e da parte imaginária da transformada Y(w)

```
> A(w):=evalc(Re(Y(w)));B(w):=evalc(Im(Y(w)));
```

Para conferirY(w)= F(w)=A(w) - iB(w)

```
> F(w):=radsimp((A(w)-I*B(w)));
```

Para o espectro

```
> modulo_Y:=evalc(abs(Y(w)));
> plot(modulo_Y,w=-4..4,thickness=2,titlefont=[COURIER,DEFAULT,10],
  labels=["w", "|Y(w|"],labelfont=[COURIER,DEFAULT,16],axesfont=
  [COURIER,DEFAULT,10],title=`|Y(w)|versus w: O Espectro de
  Amplitude`);
```

Terceiro passo: aplica-se a Transformada Inversa

```
> y(t):=invfourier(Y(w),w,t);
```

Importantíssimo: note a função erro erf no resultado. O Maple calcula diretamente a integral através de médodo numérico.

Para saber mais sobre a função erro, recorre-se ao Help do Maple, como a seguir (troque o símbolo # por ?)

```
> #erf
```

- A função erro é definida para todo x complexo por

```
erf(x) = 2/sqrt(Pi) * int(exp(-t^2), t=0..x)
```

Agora, o gráfico da solução. Observe a exponencial no denominador. para t<0 a função y(t) é decrescente e, para t>0, crescente.

```
> plot(y(t),t=-Pi..Pi,thickness=2, color=blue,scaling=constrained,
  thickness=2,titlefont=[COURIER,DEFAULT,16],labels=["t", "y(t)"],
  labelfont=[COURIER,DEFAULT,16],axesfont=[COURIER,DEFAULT,10],
  title=`Solução da EDO`);
```

O valor máximo de y (t) é calculado com limit t -> ∞ que é igual a zero; o limite em t=0 é diferente de zero e o limite para t-> - ∞ é zero,calculados a seguir

```
> limit(y(t),t=infinity);y(0):=limit(y(t),t=0);limit(y(t),t=-
```

```
infinity);
```

Para o gráfico da resposta em módulo

```
> a:=abs(y(t));plot(a,t=-Pi..Pi,thickness=2, color=green,scaling=
  constrained,titlefont=[COURIER,DEFAULT,16],labels=["t", "|y(t)
  |"],labelfont=[COURIER,DEFAULT,16],axesfont=[COURIER,DEFAULT,10],
  title=`Módulo da Solução da EDO`);
```

==

Exemplo 3

Um sistema massa-mola forçado conservativo, submetido a uma delta de Dirac.
Observe que trata-se de uma EDO de segunda ordem, então seriam necessárias duas condições iniciais. Inicialmente, será dada apenas uma condição inicial. Nestes casos a T de Fourier co-seno ou a T de Fourier seno é adequada, conforme a condição especificada. Ou seja, a primeira para a velocidade inicial e a segunda para a posição inicial.
Vamos resolver o exercício, com o comando direto, de quatro maneiras (observe as condições iniciais dadas em cada caso): 1) com a transformada Fourier co-seno; 2)com a transformada Fourier seno; 3) tentativa com o método direto da transformada Fourier usual; 4)com a transformada de Laplace.

```
> restart;with(inttrans):with(plots):
> f:=Dirac(t-3);
> assume(k>0);interface(showassumed=0);
> edo := diff(x(t),t$2) - k*x(t)=f;
> sol1:=dsolve({edo,D(x)(0)=1},x(t),method=fouriercos);solA:=
  collect(simplify(subs(k=2,sol1)),exp);
> plot(rhs(solA),t=-Pi..2*Pi,thickness=2, color=blue,titlefont=
  [COURIER,DEFAULT,16],labels=["t", "x(t)"],labelfont=[COURIER,
  DEFAULT,16],axesfont=[COURIER,DEFAULT,10],title=`Solução da EDO-T
  Fourier Co-seno`);
> sol2:=dsolve({edo,x(0)=1},x(t),method=fouriersin);solB:=collect
  (simplify(subs(k=2,sol2)),exp);
> plot(rhs(solB),t=-Pi..2*Pi,thickness=2, discont=true,color=blue,
  thickness=3,titlefont=[COURIER,DEFAULT,16],labels=["t", "x(t)"],
  labelfont=[COURIER,DEFAULT,16],axesfont=[COURIER,DEFAULT,10],
  title=`Solução da EDO-T. Fourier Seno`);
> sol3:=dsolve({edo, D(x)(0)=1,x(0)=0},x(t),method=fourier);
```

Veja que o método de Fourier, acima, não fornece solução para o problema. Entretanto, com a T de Laplace, tem-se

```
> sol4:=dsolve({edo, D(x)(0)=1,x(0)=0},x(t),method=laplace);solD:=
  subs(k=2,sol4);
```

```
> plot(rhs(solD),t=-Pi..3/2*Pi,thickness=2, color=blue,titlefont=
  [COURIER,DEFAULT,16],labels=["t", "x(t)"],labelfont=[COURIER,
  DEFAULT,16],axesfont=[COURIER,DEFAULT,10],title=`Solução da EDO-
  T. de Laplace`);
```

Terceira Propriedade: Deslocamento no eixo t

Se $\boldsymbol{F}$ {f(t)}= $\int_{-\infty}^{\infty} f(t)\, e^{-i w t}\, dt$ =F(w) e t_o é um número real, então $\boldsymbol{F}$ {f(t-t_o)}= $e^{-i w t_o}$F(w)

A transformada inversa da propriedade do deslocamento é $\boldsymbol{F}$ ^-1{F(w)}= f(t-t_o)

Exemplo 1 Determine a transformada de Fourier da função pulso de amplitude 6, ligada em t=3 e desligada em t=7, calculando antes a transformada do pulso de amplitude k, simétrico e de duração 2a

```
> restart:with(inttrans):with(plots):assume(k>0):assume(a>0)
  :interface(showassumed=0):
> f(t):=piecewise(t>3 and t<7,6);f:=convert(f(t),Heaviside);
> plot(f(t),t=0..8,thickness=2,color=blue,titlefont=[COURIER,
  DEFAULT,14],labels=["t", "f(t)"],labelfont=[COURIER,DEFAULT,16],
  axesfont=[COURIER,DEFAULT,16],title=`f(t)=  6*[Heaviside(t-3)-
  Heaviside(t-7)]`);
> g(t):=piecewise(t<-a,0,t>-a and t<a,k,t>=a,0);g:=convert(g(t),
  Heaviside);
> G(w):=fourier(g,t,w);
> g1(t):=subs((a=2,k=6),g);
> F(w):=simplify(fourier(g1(t-5),t,w));
```

Exemplo 2 Determine a transformada Fourier inversa de $\frac{e^{2 w I}}{5+I w}$

```
> restart:with(inttrans):with(plots):
> f2(t):=invfourier(exp(2*w*I)/(5+I*w),w,t);
> plot(f2(t),t=-2..1,thickness=2,color=blue,titlefont=[COURIER,
  DEFAULT,14],labels=["t", "f(t)"],labelfont=[COURIER,DEFAULT,16],
```

```
xtickmarks=[-2,-1,0,1],ytickmarks=[0,0.5,1],axesfont=[COURIER,
DEFAULT,16],title=`f(t)= exp(-5*t-10)*Heaviside(2+t)  `);
```

Quarta Propriedade: Deslocamento em freqüência

Se **F** {f(t)}= $\int_{-\infty}^{\infty} f(t)\, e^{-i w t}\, dt$ =F(w) e w_o é um número, então **F** {$e^{i w_o t}$ f(t)}= F(w-w_o)

A transformada inversa é **F** ^-1{F(w-w_o)}= $e^{i w_o t}$ f(t)

Exemplo Determine a Transformada de Fourier inversa de F(w)= $\frac{4\, e^{(2w-6)I}}{5-(3-w)I}$

```
> restart:with(inttrans):
> F(w):=4*exp((2*w-6)*I)/(5-(3-w)*I);
> f3(t):=invfourier(F(w),w,t);
```

Quinta Propriedade: Mudança de escala

Se **F** {f(t)}= $\int_{-\infty}^{\infty} f(t)\, e^{-i w t}\, dt$ =F(w) e $a \neq 0$ é um número real, então **F** (a t)}= $\frac{1}{|a|} F\left(\frac{w}{a}\right)$

A transformada inversa é **F** ^-1{$F\left(\frac{w}{a}\right)$ }= $|a|\ f(a\,t)$

Exemplo:

```
> restart:with(inttrans):with(plots):with(plottools):
> f(t):=piecewise(t <-1/2,0, t>-1/2 and t<1/2,1,t>1/2,0);
> f:=convert(f(t),Heaviside);
> g1:=plot(f,t=-2..2,thickness=2,color=blue,titlefont=[COURIER,
  DEFAULT,14],labels=["t", "f(t)"],labelfont=[COURIER,DEFAULT,16],
  axesfont=[COURIER,DEFAULT,16],title=`f(t)=Heaviside(1/2+t)-
  Heaviside(t-1/2)  `):g1;
> F(w):=fourier(f,t,w);
> modulo_F:=abs(F(w));
```

Cálculo dos nós

```
> with(Student[Calculus1]):
  r:=Roots(modulo_F,-10..10);pontos:=evalf(r);
```

Para testar, vamos calcular estes limites para |F(w)| dados k=1, e a=1/2

```
>  a:=1/2;nos_wpos:=seq(limit(modulo_F,w=n*Pi/a),n=-3..3);
> G1:=plot(modulo_F,w=-10..10,thickness=2,color=blue,titlefont=
  [COURIER,DEFAULT,16],labels=["w", "|F(w)|"],labelfont=[COURIER,
  DEFAULT,16],ytickmarks=[0,1,2,3,4],xtickmarks=[-9.3,-6.3,-3.1,0,
  3.1,6.2,9.3],axesfont=[COURIER,DEFAULT,16],title=`|F(w)|versus W:
```

```
O Espectro de Amplitude`):G1;
```

Agora vamos calcular as transformadas de f(2 t) e de f(t/2)

Primeiro: a transformada de f(2*t)

```
> f2:=subs(t=2*t,f);
```

```
> g2:=plot(f2,t=-2..2,thickness=2,color=green,titlefont=[COURIER,
  DEFAULT,14],labels=["t", "f(t)"],labelfont=[COURIER,DEFAULT,16],
  axesfont=[COURIER,DEFAULT,16],title=`f(t)=Heaviside(1/2+2t)-
  Heaviside(2t-1/2)  `):g2;
```

```
> F2(w):=fourier(f2,t,w);
```

```
> modulo_F2:=abs(F2(w));
```

Cálculo dos nós

```
> with(Student[Calculus1]):
  r:=Roots(modulo_F2,-20..20);pontos:=evalf(r);
```

Para testar, vamos calcular estes limites para |F(w)| dados k=1, e a=1/4

```
>  a:=1/4;nos_wpos:=seq(limit(modulo_F2,w=n*Pi/a),n=-3..3);
```

```
> G2:=plot(modulo_F2,w=-20..20,thickness=2,color=green,titlefont=
  [COURIER,DEFAULT,16],labels=["w", "|F(w)|"],labelfont=[COURIER,
  DEFAULT,16],ytickmarks=[0,0.5],xtickmarks=[-12.56637062,
  12.56637062],axesfont=[COURIER,DEFAULT,16],title=` O Espectro de
  Amplitude`):G2;
```

Segundo: a transformada de f(t/2)

```
> f3:=subs(t=t/2,f);
```

```
> g3:=plot(f3,t=-2..2,thickness=2,color=orange,titlefont=[COURIER,
  DEFAULT,14],labels=["t", "f(t)"],labelfont=[COURIER,DEFAULT,16],
  axesfont=[COURIER,DEFAULT,16],title=`f(t)=Heaviside(1/2+t/2)-
  Heaviside(t/2-1/2)  `):g3;
```

```
> F3(w):=fourier(f3,t,w);
```

```
> modulo_F3:=abs(F3(w));
```

Cálculo dos nós

```
> with(Student[Calculus1]):
  r:=Roots(modulo_F3,-10..10);Digits:=3;pontos:=evalf(r);
```

Para testar, vamos calcular estes limites para |F(w)| dados k=1, e a=1/2

```
>  a:=1;nos_wpos:=seq(limit(modulo_F3,w=n*Pi/a),n=-3..3);
```

```
> G3:=plot(modulo_F3,w=-10..10,color=orange,thickness=2,titlefont=
  [COURIER,DEFAULT,16],labels=["w", "|F(w)|"],labelfont=[COURIER,
  DEFAULT,10],ytickmarks=[0,1,2],xtickmarks=[-9.42, -6.28, -3.14,
  3.14, 6.28, 9.42],axesfont=[COURIER,DEFAULT,8],title=` O Espectro
  de Amplitude`):G3;
```

```
> display({g1,g2,g3},title=`Pulsos`);
> display({G1,G2,G3},title=`Espectros`);
```

Comparar os gráficos acima e associar o pulso com o respectivo (de mesma cor) espectro de amplitude.
O quê se pode inferir?

Sexta Propriedade: Inversão temporal

$$Se\ \boldsymbol{F}\ \{f(t)\}= \int_{-\infty}^{\infty} f(t)\, e^{-i w t}\, dt = F(w)\ \text{, então } \boldsymbol{F}\ \{f(-t)\}=F(-w)$$

```
> restart:with(inttrans):with(plots):
> f(t):=piecewise(t <1,0, t>1 and t<2,1,t>2,0);
> f:=convert(f(t),Heaviside);
> g1:=plot(f,t=-4..4,thickness=2,color=blue,titlefont=[COURIER,
  DEFAULT,14],labels=["t", "f(t)"],labelfont=[COURIER,DEFAULT,16],
  axesfont=[COURIER,DEFAULT,16],title=`f(t)=Heaviside(-1+t)-
  Heaviside(t-2)  `):g1;
> F(w):=combine(fourier(f,t,w),exp);
> f2:=subs(t=-t,f);
> g2:=plot(f2,t=-3..3,thickness=2,color=green,titlefont=[COURIER,
  DEFAULT,14],labels=["t", "f(t)"],labelfont=[COURIER,DEFAULT,16],
  axesfont=[COURIER,DEFAULT,16],title=`f(t)=Heaviside(-1-t)-
  Heaviside(-t-2)  `):g2;
> F2(w):=fourier(f2,t,w);
```

Sétima Propriedade: Simetria ou dualidade

$$Se\ \boldsymbol{F}\ \{f(t)\}= \int_{-\infty}^{\infty} f(t)\, e^{-i w t}\, dt = F(w)\ \text{, então } \boldsymbol{F}\ \{F(t)\}= 2\pi f(-w)$$

EXEMPLO1

```
> restart:with(inttrans):
> f(t):=piecewise(t <-1/2,0, t>-1/2 and t<1/2,1,t>1/2,0);
> f:=convert(f(t),Heaviside);
> F(w):=fourier(f,t,w);
> F(t):=subs(w=t,F(w));
> f2:=fourier(F(t),t,w);
```

EXEMPLO 2 Dada f(t):=5/(4+I*t), mostrar que $\boldsymbol{F}$ {F(t)}= 2πf(-w)

```
> restart:with(inttrans):

> f(t):=5/(4+I*t);
```

```
> G:=subs(t=w,f(t)) ;
> transf_G:=invfourier(G,w,t);
> f2:=fourier(5/(4+t*I),t,w);
```

Oitava Propriedade: Teorema da Modulação

Se $\boldsymbol{F}$ {f(t)}= $\int_{-\infty}^{\infty} f(t)\, e^{-i w t}\, dt$ =F(w) e w_o é um número real , então

$$\boldsymbol{F}\{f(t)\cos(w_o t)\} = \frac{1}{2}[\,F(w-w_o) + F(w+w_o)]$$

$$\boldsymbol{F}\{f(t)\sin(w_o t)\} = \frac{I}{2}[\,F(w+w_o) - F(w-w_o)]$$

```
> restart:with(inttrans):with(plots):assume(a>0);assume(k>0);
  interface(showassumed=0):
```

Dada a função pulso $f(t) := \begin{cases} k\sim & -a\sim - t < 0 \textbf{ and } t - a\sim < 0 \\ 0 & a\sim + t < 0 \textbf{ and } -t + a\sim < 0 \end{cases}$, determinar a transformada do pulso modulado g(t)=f(t)*cos(2*t)

```
> f(t):=piecewise(t>-a and t<a, k, t<-a and t>a,0) ;
> f:=convert(f(t),Heaviside);
```

A F(w) inicial é

```
> F:=fourier(f,t,w);
> a:=2;g(t):=piecewise(t>-a and t<a, 2*cos(2*t), t<-a and t>a,0);

> plot(g(t),t=-5..4,discont=true,thickness=3,scaling=constrained,
  color=blue,titlefont=[COURIER,DEFAULT,16],labels=["t", "g(t)"],
  labelfont=[COURIER,DEFAULT,16],axesfont=[COURIER,DEFAULT,10],
  title=`Pulso-Modulado`);
```

Primeiro vamos calcular a transformada de Fourier do pulso-modulado (escrito como uma função deHeaviside) com o comando do Maple para a transformada de Fourier

```
> g1(t):=convert(g(t),Heaviside);
> G(w):=combine(fourier(g1(t),t,w),trig);
> r1:=collect(G(w),[sin(2*w+4),sin(2*w-4)]);
```

Para conferir com a Propriedade da Modulação, consideraremos a F(w) inicial:

```
> F := 2*k/w*sin(w*a);G1(W):=((1/2)*subs((k=2,a=2,w=W+2),F(w)));G2
  (W):=((1/2)*subs((k=2,a=2,w=W-2),F(w)));G(w):=G1(W)+G2(W);

> r2:=collect((G(w)),[sin(2*W+4),sin(2*W-4)]);
```

Agora, compare os resultados r1 e r2.

O Espectro

```
> espec_G:=abs(G(w));
```

```
> plot(espec_G,W=-10..10,scaling=constrained,discont=true,
  thickness=3,color=red,titlefont=[COURIER,DEFAULT,16],labels=["w",
  "|G(w)|"],labelfont=[COURIER,DEFAULT,16],axesfont=[COURIER,
  DEFAULT,10],title=`Espectro do Pulso-Modulado`);
```

===